In[1]:=
```
ClearAll["Global`*"]
```

In[2]:=
```
k = 4
```

Out[2]= 4

This code attempts computing analytically the expectations for the various terms in Eq. 28 (Sec B1) of the Supplementary Information. The code computes the expectations for n = {2,3,...k} (with k small enough), and uses these values to verify the equations in Sec B1. To verify the equations for larger values of n, increase the value k above (the computing time grows exponentially; the values have been verified for n = 10)

Outer product

In[3]:=
```
Op[a_, b_] := Outer[Times, a, b]
```

Build skew-symmetric matrix with arbitrary coefficients

In[4]:=
```
BuildB[n_] := Module[{A, B}, A = UpperTriangularize[Table[b[i, j], {i, 1, n}, {j, 1, n}], 1];
  B = A - Transpose[A];
  B]
```

Remove the mean from matrix B

In[5]:=
```
BuildC[n_] := Module[{B, ones, m, K},
  B = BuildB[n];
  ones = Table[1, n];
  m = 1/n Flatten[Transpose[B].ones]; K = B - Op[ones, m]; K]
```

This is the right hand side of the equations

In[6]:=
```
Buildb[n_] := Table[1/α, n]
```

Build the initial guess

In[7]:=
```
Buildy0[n_] := Table[1/α, n] - 1/α^2 BuildC[n].Table[1, n]
```

This is the matrix M (which depends on the correlation; this is the version for skew symmetric matrices, as specified in Sec B1)

In[8]:=
```
BuildM[n_] := Module[{M},
  M = (IdentityMatrix[n] + 1/α Transpose[BuildC[n]]).(IdentityMatrix[n] + 1/α BuildC[n]); M]
```

These are the expectations for the powers of Bij

In[9]:=
```
r6 := Flatten[Table[b[i, j]^6 → μ6, {i, 1, n}, {j, 1, n}]]
```

In[10]:=
```
r5 := Flatten[Table[b[i, j]^5 → μ5, {i, 1, n}, {j, 1, n}]]
```

In[11]:=
```
r4 := Flatten[Table[b[i, j]^4 → μ4, {i, 1, n}, {j, 1, n}]]
```

```
In[12]:= r3 := Flatten[Table[b[i, j]^3 → μ3, {i, 1, n}, {j, 1, n}]]

In[13]:= r2 := Flatten[Table[b[i, j]^2 → μ2, {i, 1, n}, {j, 1, n}]]

In[14]:= rrem := Flatten[Table[b[i, j] → 0, {i, 1, n}, {j, 1, n}]]
```

Compute expectation given a matrix of expressions

```
In[15]:= ExpandMat[Z_] := Total[Total[ExpandAll[Z] /. r6 /. r5 /. r4 /. r3 /. r3 /. r2 /. r2 /. r2 /. rrem]]

In[16]:= ExpandExpression[x_] := ExpandAll[x] /. r6 /. r5 /. r4 /. r3 /. r3 /. r2 /. r2 /. r2 /. rrem
```

Having set up the problem, we compute all the terms in Eq 28

1. Term b^T b. First, we compute the expectation for this term for different values of n

```
In[17]:= btb = FullSimplify[Table[Buildb[n].Buildb[n], {n, 2, k}]]
```

Out[17]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

```
In[18]:= predictedbtb = Table[n / α ^ 2, {n, 2, k}]
```

Out[18]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

```
In[19]:= FullSimplify[btb - predictedbtb]
```

Out[19]=

```
{0, 0, 0}
```

2. Term b^T y_0. First, we compute the expectation for this term for different values of n

```
In[20]:= bty0 = FullSimplify[Table[Buildb[n].Buildy0[n], {n, 2, k}]]
```

Out[20]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

```
In[21]:= predictedbty0 = Table[n / α ^ 2, {n, 2, k}]
```

Out[21]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}, \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

In[22]:=
```
FullSimplify[bty0 - predictedbty0]
```
Out[22]=

{0, 0, 0}

3. Term $y_0^T\, y_0$. First, we compute the expectation for this term for different values of n

In[23]:=
```
y0ty0 = FullSimplify[Table[ExpandExpression[Buildy0[n].Buildy0[n]], {n, 2, k}]]
```
Out[23]=

$$\left\{\frac{2\,(\alpha^2+\mu 2)}{\alpha^4},\ \frac{3\,(\alpha^2+2\,\mu 2)}{\alpha^4},\ \frac{4\,(\alpha^2+3\,\mu 2)}{\alpha^4}\right\}$$

The formula in Sec B1 reads

In[24]:=
```
predictedy0ty0 = Table[n / α ^ 2 + 1 / α ^ 4 μ2 (n - 1) (n), {n, 2, k}]
```
Out[24]=

$$\left\{\frac{2}{\alpha^2}+\frac{2\,\mu 2}{\alpha^4},\ \frac{3}{\alpha^2}+\frac{6\,\mu 2}{\alpha^4},\ \frac{4}{\alpha^2}+\frac{12\,\mu 2}{\alpha^4}\right\}$$

Make sure that the terms match

In[25]:=
```
FullSimplify[y0ty0 - predictedy0ty0]
```
Out[25]=

{0, 0, 0}

4. Term $b^T\, M\, y_0$. First, we compute the expectation for this term for different values of n

In[26]:=
```
btMy0 = Table[ExpandExpression[FullSimplify[Buildb[n].BuildM[n].Buildy0[n]]], {n, 2, k}]
```
Out[26]=

$$\left\{\frac{2}{\alpha^2},\ \frac{3}{\alpha^2},\ \frac{4}{\alpha^2}\right\}$$

The formula in Sec B1 reads

In[27]:=
```
predictedbtMy0 = Table[n / α ^ 2, {n, 2, k}]
```
Out[27]=

$$\left\{\frac{2}{\alpha^2},\ \frac{3}{\alpha^2},\ \frac{4}{\alpha^2}\right\}$$

Make sure that the terms match

In[28]:=
```
FullSimplify[btMy0 - predictedbtMy0]
```
Out[28]=

{0, 0, 0}

5. Term $y_0^T\, M\, y_0$. First, we compute the expectation for this term for different values of n

In[29]:=
```
y0tMy0 = Table[ExpandExpression[FullSimplify[Buildy0[n].BuildM[n].Buildy0[n]]], {n, 2, k}]
```

Out[29]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}+\frac{2\,\mu4}{\alpha^6}, \frac{4}{\alpha^2}+\frac{6\,\mu2^2}{\alpha^6}+\frac{6\,\mu4}{\alpha^6}\right\}$$

The formula in Sec B1 reads

In[30]:=
```
predictedy0tMy0 = Table[(n) / α ^ 2 + (n - 1) (n - 2) μ4 / α ^ 6 + (n - 1) (n - 2) (n - 3) μ2 ^ 2 / α ^ 6, {n, 2, k}]
```

Out[30]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}+\frac{2\,\mu4}{\alpha^6}, \frac{4}{\alpha^2}+\frac{6\,\mu2^2}{\alpha^6}+\frac{6\,\mu4}{\alpha^6}\right\}$$

Make sure that the terms match

In[31]:=
```
FullSimplify[y0tMy0 - predictedy0tMy0]
```

Out[31]=
```
{0, 0, 0}
```

6. Term $y_0^T M^T M y_0$. First, we compute the expectation for this term for different values of n

In[32]:=
```
y0tMtMy0 = Table[ExpandExpression[
   FullSimplify[Buildy0[n].Transpose[BuildM[n]].BuildM[n].Buildy0[n]]], {n, 2, k}]
```

Out[32]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}-\frac{4\,\mu2^3}{\alpha^8}-\frac{4\,\mu3^2}{9\,\alpha^8}+\frac{2\,\mu4}{\alpha^6}+\frac{4\,\mu2\,\mu4}{\alpha^8}+\frac{2\,\mu6}{3\,\alpha^8}, \frac{4}{\alpha^2}+\frac{6\,\mu2^2}{\alpha^6}+\frac{12\,\mu2^3}{\alpha^8}-\frac{2\,\mu3^2}{\alpha^8}+\frac{6\,\mu4}{\alpha^6}+\frac{21\,\mu2\,\mu4}{\alpha^8}+\frac{3\,\mu6}{\alpha^8}\right\}$$

The formula in Sec B1 reads

In[33]:=
```
predictedy0tMtMy0 =
 Table[n / α ^ 2 + μ6 (n - 1) (n - 2) ^ 2 / (n α ^ 8) + μ2 ^ 3 (n - 2) (n - 1) (-60 + 45 n - 15 n^2 + 2 n^3) / (n α ^ 8) +
   μ2 ^ 2 (n - 1) (n - 2) (n - 3) / α ^ 6 + μ2 μ4 (2 (-2 + n) (-1 + n) (15 - 10 n + 2 n^2)) / (n α ^ 8) +
   μ4 (n - 1) (n - 2) / α ^ 6 + μ3 ^ 2 2 / 3 (n - 1) (n - 2) (n ^ 2 - 8 n + 14) / (n α ^ 8), {n, 2, k}]
```

Out[33]=

$$\left\{\frac{2}{\alpha^2}, \frac{3}{\alpha^2}-\frac{4\,\mu2^3}{\alpha^8}-\frac{4\,\mu3^2}{9\,\alpha^8}+\frac{2\,\mu4}{\alpha^6}+\frac{4\,\mu2\,\mu4}{\alpha^8}+\frac{2\,\mu6}{3\,\alpha^8}, \frac{4}{\alpha^2}+\frac{6\,\mu2^2}{\alpha^6}+\frac{12\,\mu2^3}{\alpha^8}-\frac{2\,\mu3^2}{\alpha^8}+\frac{6\,\mu4}{\alpha^6}+\frac{21\,\mu2\,\mu4}{\alpha^8}+\frac{3\,\mu6}{\alpha^8}\right\}$$

Make sure that the terms match

In[34]:=
```
FullSimplify[y0tMtMy0 - predictedy0tMtMy0]
```

Out[34]=
```
{0, 0, 0}
```